

BATNUM
(BATTLE OF NUMBERS)

```

5.
6.
7.
8:RETURN
9. BATNUM
10. CREATIVE COMPUTING
11. FOR BALLY ARCADE
12. BY RON SCHWENK
20 NT=1;CLEAR;BC=14;FC=8
40 PRINT ;PRINT "          BATNUM
50 PRINT ;PRINT "IF YOU WANT INSTRUCTIONS,
60 PRINT "PUSH 'GO'."
70 K=KP
80 IF K#13GOTO 320
100 CLEAR
110 PRINT "THIS PROGRAM IS A BATTLE
120 PRINT "OF NUMBERS WHERE THE
130 PRINT "COMPUTER IS YOUR OPPONENT.
150 PRINT "THE GAME STARTS WITH AN
160 PRINT "ASSUMED PILE OF OBJECTS.
170 PRINT "YOU AND THE COMPUTER
180 PRINT "ALTERNATELY REMOVE OBJECTS
190 PRINT "FROM THE PILE. WINNING
200 PRINT "IS DETERMINED BY YOU AT THE
210 PRINT "START AS TAKING THE LAST
220 PRINT "OBJECT OR NOT. DON'T PICK
230 PRINT "ZERO FOR YOUR MOVE.
240 PRINT "PUSH 'GO' TO START GAME.",
260 K=KP
270 IF K#13GOTO 260
320 CLEAR ;BC=112;FC=124
330 INPUT "ENTER PILE SIZE"N
340 IF N#0 GOTO 370
350 GOTO 330
370 PRINT "ENTER WIN OPTION
380 PRINT " 1 - TO TAKE LAST
390 PRINT " 2 - TO AVOID LAST
400 CX=0;INPUT "?M
410 IF M=1GOTO 430
420 IF M#2GOTO 370
430 INPUT "MINIMUM DRAW?"A
435 INPUT "MAXIMUM?"B
440 IF A>BGOTO 430
445 IF A=BGOTO 430
450 IF A<1GOTO 430
480 PRINT "ENTER START OPTION
490 PRINT " 1 - COMPUTER FIRST
500 PRINT " 2 - YOU FIRST
510 CX=0;INPUT "?S
515 IF S=1GOTO 530
520 IF S#2GOTO 480
530 C=A+B
532 CLEAR;FC=5;BC=176
535 IF M=1PRINT "TAKE",
536 IF M=2PRINT "AVOID",
537 PRINT " LAST OF ",#1,N," ("A,"-",B,")"
540 IF S=2GOTO 570
550 GOSUB 600
560 IF W=1GOTO 1100
570 GOSUB 810
580 IF W=1GOTO 1100
590 GOTO 550
600 Q=N
610 IF M=1GOTO 630
620 Q=Q-1
630 IF M=1GOTO 680
640 IF N>AGOTO 720
650 W=1
660 PRINT "COMPUTER TAKES ",#1,N," AND
        LOSES"
670 RETURN
680 IF N>BGOTO 720
690 W=1
700 PRINT "COMPUTER TAKES ",#1,N," AND
        WINS"
710 RETURN
720 P=Q-CX(Q+C)
730 IF P>=AGOTO 750
740 P=A
750 IF P>=BGOTO 770
760 P=B
770 N=N-P
780 PRINT "I TAKE ",#1,P," LEAVING ",N
790 W=0
800 RETURN
810 INPUT "          YOUR MOVE"P
830 IF P#0GOTO 880
840 PRINT "I TOLD YOU NOT TO USE ZERO
845 PRINT "SO YOU FORFEIT THE GAME!"
850 W=1
860 RETURN
880 IF P>=AGOTO 910
890 IF P=NGOTO 960
900 GOTO 920
910 IF P<=BGOTO 940
920 PRINT "ILLEGAL MOVE
930 GOTO 810
940 N=N-P
950 IF N#0GOTO 1030
960 IF M=1GOTO 1000
970 PRINT "TOUGH LUCK, YOU LOSE."
980 W=1;RETURN
1000 PRINT "CONGRATULATIONS WINNER
1010 W=1;RETURN
1030 IF N>=0GOTO 1060
1040 N=N+P
1050 GOTO 920
1060 W=0;RETURN
1100 PRINT
1110 PRINT "PUSH 'GO' TO PLAY AGAIN
1120 K=KP
1130 IF K=13 GOTO 320
1140 PRINT;PRINT "OK. HOPE TO TRY BATNUM
1150 PRINT "AGAIN SOON."

```